
1. Overview

This document lists TTCN changes needed to correct issues in the ATS ‘iwd-TTCN3-B2017-06_D17wk37’.

2. Corrections required

2.1 Change 1 – Correction to function ‘f_TC_22_3_1_4_NBIOT()’

	Function name
	f_TC_22_3_1_4_NBIOT()

	Reason for change
	1. After step 5 UE will transmit a random access preamble(refering to TS36.321 clause 5.4.5) and send BSR & CRNTI & part of data(first RLC SDU) in msg3 if received RAR, new steps 5A-5C are needed. In order to empty the UL RLC buffer and prepare for next Regular BSR triggered on UE side, change steps 6-7 to allocate UL grant(msg 4) and receive the remain part of data(RLC SDUs sent at step 2),  then send RLC STATUS at new step 7A.
2. Receiving handling for step 5C is needed.
3. Since msg 4(UL Grant) at step 6 is automatically sent by SS, restarting the timing at step 8 is needed, adjust timing at step 9, and change BSR Buffer Size at step 10 accordingly. 
4. In order to empty the UL RLC buffer and prepare for next Regular BSR triggered on UE side before steps 11-13, add new steps 10A-10C to allocate UL grant, receive the data(RLC SDUs sent at step 8) then send a RLC STATUS.
5. Step 11 waiting for retxBSR-Timer expiry can not satisfy the trigger condition of Test Purpose 4, instead, sending a RLC SDU triggers a Regular BSR.
6. Timing is now for Step 20, it maybe too short to handle the loopbacked data by UE and cause step 21 fail. Restarting the timing at step 19 and adjusting timing at step 20 are needed.
7. In order to eliminate interactive effects between retxBSR-Timer expiry and periodicBSR-Timer expiry, change the value of retxBSR-Timer > periodicBSR-Timer when testing Regular BSR, accordingly, change the value of periodicBSR-Timer > retxBSR-Timer when testing periodicBSR-Timer.
8. Considering above retxBSR-Timer and periodicBSR-Timer are changed, new steps 21B-21C to reconfigure the value of retxBSR-Timer and periodicBSR-Timer in RRCConnectionSetup-NB are needed.

	Summary of change
	1. Add new steps 5A-5C, modify steps 6-7 and add new step 7A according to the associated Prose CR. 
2. Add new function fl_TC_22_3_1_4_MacPdu_Receive_step5C and fl_TC_22_3_1_4_MacPdu_CRNTI_BSR_MacSdus for step 5C to receive msg3 including BSR, CRNTI and portion data of RLC UL SDU. 
3. Change timing at step 8/9 and BSR Buffer Size at step 10 according to the associated Prose CR.
4. Add new steps 10A-10C according to the associated Prose CR.
5. Change step 11 to send 3 RLC SDUs and the timing at step 12 according to the associated Prose CR.
6. Restart the timing at step 19 and adjust timing at step 20.
7. Change v_RetxBSR_Timer to pp16, v_PeriodicBSR_Timer to pp64 in RRCConnectionSetup-NB for step 1-21B; add new variable v_RetxBSR_Timer _Step21C = pp=64, v_PeriodicBSR_Timer _Step21C = pp16 in RRCConnectionSetup-NB for step 21C-27 according to the associated Prose CR.
8. Add new steps 21B-21C according to the associated Prose CR.
Associated Prose CR will be submitted in next RAN5 meeting.

	TTCN module
	NBIOT_MAC_Testcases.ttcn

	MCC160 Comment
	


Before change

  function f_TC_22_3_1_4_NBIOT() runs on NBIOT_PTC

  {

    var NBIOT_CellId_Type v_CellId := nbiot_Cell1;

    var NBIOT_IDLEUPDATED_STATE_Type v_TestLoopMode := STATE2A_NB_TESTLOOP_ModeG;      // !!!! FFS: to be specified in the prose !!!!

    var RetxBSR_Timer_NB_r13 v_RetxBSR_Timer := pp4;

    var PeriodicBSR_Timer_NB_r13 v_PeriodicBSR_Timer := pp4;

    var template (value) MAC_MainConfig_NB_r13 v_MAC_MainConfig := cs_MAC_MainConfig_NB_SRB;

    var template (value) DL_CCCH_Message_NB v_RrcConnSetup;

    var NB_RLC_CountsInfoList_Type v_RLC_CountsInfoListByRef;                        // RLC status to be maintained by Send/Receive functions

    var SubFrameTiming_Type v_SubFrameTiming;

    var template (value) TimingInfo_Type v_TimingInfo;

    var template (value) RLC_SDUList_Type v_RlcSduList;

    var template (present) MAC_SDUList_Type v_MAC_SDUList_RX;

    var template (present) MAC_PDU_Type v_MAC_PDU_RX;

    var NB_L2_DATA_IND v_NB_L2_DATA_IND;

    var integer v_NoOfNpdcchPeriods;

    var integer v_BSR_Index_Min;

    var integer i;

    v_MAC_MainConfig.ul_SCH_Config_r13.periodicBSR_Timer_r13 := v_PeriodicBSR_Timer;

    v_MAC_MainConfig.ul_SCH_Config_r13.retxBSR_Timer_r13 := v_RetxBSR_Timer;

    v_RrcConnSetup := f_NBIOT_508_RRCConnectionSetup(-, -, v_MAC_MainConfig);

    f_NBIOT_L2_Preamble_State2B_NB(v_RLC_CountsInfoListByRef, v_CellId, TransparentMode_RLC_MAC, v_TestLoopMode, v_RrcConnSetup);

    //------------------------------------

    f_NBIOT_TestBody_Set(true, layer2);

    //@siclog "Step 1 (void)" siclog@

    //@siclog "Step 2" siclog@

    v_SubFrameTiming := f_NBIOT_GetNextSearchSpace(v_CellId, 200);    /* @sic R5s170238 change 2.1: 200ms offset sic@ */

    v_TimingInfo := cs_TimingInfo_SubFrameFull(v_SubFrameTiming);

    v_RlcSduList := { crs_PDCP_SDU_10B, crs_PDCP_SDU_10B };

    f_NBIOT_MAC_CPMode_SendRlcSdus(v_RLC_CountsInfoListByRef, v_CellId, v_TestLoopMode, v_TimingInfo, v_RlcSduList);

    v_BSR_Index_Min := 6;

    //@siclog "Step 3 - 4" siclog@

    v_SubFrameTiming := f_NBIOT_IncrementSearchSpace(v_SubFrameTiming, 3.0);     /* @sic R5-172835/R5s170238 change 2.1: three NPDCCH cycles to give the UE time for processing the data sic@ */

    fl_TC_22_3_1_4_TriggerBSR(v_RLC_CountsInfoListByRef, v_CellId, v_SubFrameTiming, v_BSR_Index_Min);

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 4");

    //@siclog "Step 5" siclog@

    v_NoOfNpdcchPeriods := fl_TC_22_3_1_4_RetxBSR_Timer_NpdcchPeriods(v_RetxBSR_Timer) + 1;  /* "+1" to be sure that RetxBSR-Timer has been expired */

    v_SubFrameTiming := f_NBIOT_IncrementSearchSpace(v_SubFrameTiming, int2float(v_NoOfNpdcchPeriods));
    //@siclog "Step 6 - 7" siclog@

    fl_TC_22_3_1_4_TriggerBSR(v_RLC_CountsInfoListByRef, v_CellId, v_SubFrameTiming, v_BSR_Index_Min);

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 7");
    //@siclog "Step 8" siclog@

    v_SubFrameTiming := f_NBIOT_IncrementSearchSpace(v_SubFrameTiming, 3.0);    /* three NPDCCH cycles after BSR transmission/receiption
                                                                                   @sic R5-172835/R5s170238 change 2.2: additional 2 NPDCCH cycles to give the SS enough time to build up the DL PDU sic@ */

    v_TimingInfo := cs_TimingInfo_SubFrameFull(v_SubFrameTiming);

    v_RlcSduList := { crs_PDCP_SDU_10B };

    f_NBIOT_MAC_CPMode_SendRlcSdus(v_RLC_CountsInfoListByRef, v_CellId, v_TestLoopMode, v_TimingInfo, v_RlcSduList);

    //@siclog "Step 9 - 10" siclog@

    v_SubFrameTiming := f_NBIOT_IncrementSearchSpace(v_SubFrameTiming, 2.0);    /* two NPDCCH cycles after sending of the additional RLC SDU at step 8 @sic R5-172835/R5s170238 change 2.3: two cycles instead of one sic@ */

    v_BSR_Index_Min := 8;

    fl_TC_22_3_1_4_TriggerBSR(v_RLC_CountsInfoListByRef, v_CellId, v_SubFrameTiming, v_BSR_Index_Min);

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 10");

    //@siclog "Step 11" siclog@

    v_SubFrameTiming := f_NBIOT_IncrementSearchSpace(v_SubFrameTiming, int2float(v_NoOfNpdcchPeriods));
    //@siclog "Step 12" siclog@

    v_TimingInfo := cs_TimingInfo_SubFrameFull(v_SubFrameTiming);

    f_NBIOT_ULGrantTransmission_ExplicitGrant(v_CellId, 1, 9, v_TimingInfo);  // UL grant of 296 bits (ITBS=9, IRU=1) -> 37 bytes

    //@siclog "Step 13" siclog@

    v_RlcSduList := { crs_PDCP_SDU_10B, crs_PDCP_SDU_10B, crs_PDCP_SDU_10B };

    v_MAC_SDUList_RX := { f_NBIOT_MacSdu_RlcAmdPdu_RX(v_RlcSduList) };      /* 2 bytes RLC AM header fixed part, 3 bytes for 2 LIs, 3 * 10 bytes data */

    v_MAC_PDU_RX := f_NBIOT_MacPdu_BSR_MacSdus_Padding_RX(-, singleBytePadding, noBSR, v_MAC_SDUList_RX);  /* 1 byte padding, 1 byte SDU sub-header, 35 bytes MAC SDU */

    f_NBIOT_MAC_CPMode_MacPdu_Receive(v_RLC_CountsInfoListByRef, v_CellId, v_MAC_PDU_RX);

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 13");

    //@siclog "Step 14" siclog@

    f_NBIOT_MAC_CPMode_SendRlcStatus(v_RLC_CountsInfoListByRef, v_CellId);

    //@siclog "Step 15" siclog@

    v_SubFrameTiming := f_NBIOT_GetNextSearchSpace(v_CellId, 200);    /* @sic R5s170238 change 2.5: 200ms offset sic@ */

    v_TimingInfo := cs_TimingInfo_SubFrameFull(v_SubFrameTiming);

    v_RlcSduList := { crs_PDCP_SDU_8B, crs_PDCP_SDU_7B };

    f_NBIOT_MAC_CPMode_SendRlcSdus(v_RLC_CountsInfoListByRef, v_CellId, v_TestLoopMode, v_TimingInfo, v_RlcSduList);

    //@siclog "Step 16" siclog@

    v_SubFrameTiming := f_NBIOT_IncrementSearchSpace(v_SubFrameTiming, 3.0);     /* @sic R5-172835/R5s170238 change 2.5: three NPDCCH cycles to give the UE time for processing the data sic@ */

    v_TimingInfo := cs_TimingInfo_SubFrameFull(v_SubFrameTiming);

    f_NBIOT_ULGrantTransmission_ExplicitGrant(v_CellId, 1, 8, v_TimingInfo);     /* UL grant of 256 bits (ITBS=8, IRU=1)  @sic R5s170238 change 2.6: ITBS=8 sic@ */

    //@siclog "Step 17" siclog@

    v_MAC_SDUList_RX := { f_NBIOT_MacSdu_RlcAmdPdu_RX(v_RlcSduList) };      /* 2 bytes RLC AM header fixed part, 2 bytes for 1 LI, 8 + 7 bytes data */
    v_MAC_PDU_RX := f_NBIOT_MacPdu_BSR_MacSdus_Padding_RX(-, normalPadding, zeroBSR, v_MAC_SDUList_RX);

    f_NBIOT_MAC_CPMode_MacPdu_Receive(v_RLC_CountsInfoListByRef, v_CellId, v_MAC_PDU_RX);

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 17");

    //@siclog "Step 18" siclog@

    f_NBIOT_MAC_CPMode_SendRlcStatus(v_RLC_CountsInfoListByRef, v_CellId);

    //@siclog "Step 19" siclog@

    v_RlcSduList := { crs_PDCP_SDU_12B };

    f_NBIOT_MAC_CPMode_SendRlcSdus(v_RLC_CountsInfoListByRef, v_CellId, v_TestLoopMode, -, v_RlcSduList);

    //@siclog "Step 20" siclog@

    f_NBIOT_ULGrantTransmission_ExplicitGrant(v_CellId, 0, 9);  // UL grant of 136 bits (ITBS=9, IRU=0)

    //@siclog "Step 21" siclog@

    v_MAC_SDUList_RX := { f_NBIOT_MacSdu_RlcAmdPdu_RX(v_RlcSduList) };

    v_MAC_PDU_RX := f_NBIOT_MacPdu_BSR_MacSdus_Padding_RX(-, noPadding, zeroBSR, v_MAC_SDUList_RX);            /* @sic R5s170238 change 2.8 - MMC160 implementation: zeroBSR instead of anyBSR sic@ */

    v_NB_L2_DATA_IND := f_NBIOT_MAC_CPMode_MacPdu_Receive(v_RLC_CountsInfoListByRef, v_CellId, v_MAC_PDU_RX);

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 21");

    //@siclog "Step 21A" siclog@

    f_NBIOT_MAC_CPMode_SendRlcStatus(v_RLC_CountsInfoListByRef, v_CellId);                                     /* @sic R5-172835/R5s170238 change 2.9 sic@ */

    //@siclog "Step 22" siclog@

    v_RlcSduList := { crs_PDCP_SDU_14B };

    f_NBIOT_MAC_CPMode_SendRlcSdus(v_RLC_CountsInfoListByRef, v_CellId, v_TestLoopMode, -, v_RlcSduList);

    //@siclog "Step 23 - 24" siclog@

    v_SubFrameTiming := f_NBIOT_GetNextSearchSpace(v_CellId);

    v_BSR_Index_Min := 1;

    fl_TC_22_3_1_4_TriggerBSR(v_RLC_CountsInfoListByRef, v_CellId, v_SubFrameTiming, v_BSR_Index_Min);

    for (i := 0; i < 2; i := i + 1) {

      //@siclog "Step 25" siclog@

      v_BSR_Index_Min := 1;

      v_NoOfNpdcchPeriods := fl_TC_22_3_1_4_PeriodicBSR_Timer_NpdcchPeriods(v_PeriodicBSR_Timer) + 1;  // "+1" to be sure that PeriodicBSR-Timer has been expired

      v_SubFrameTiming := f_NBIOT_IncrementSearchSpace(v_SubFrameTiming, int2float(v_NoOfNpdcchPeriods));

      //@siclog "Step 26 - 27" siclog@

      fl_TC_22_3_1_4_TriggerBSR(v_RLC_CountsInfoListByRef, v_CellId, v_SubFrameTiming, v_BSR_Index_Min);

      f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 27 - " & int2str(i));

    }

    f_NBIOT_TestBody_Set(false);

    //------------------------------------

    // give default grant to the UE to allow transmission of pending data

    f_NBIOT_ULGrantTransmission(v_CellId, cs_NB_UL_GrantScheduling_Start(cs_NB_SingleGrant));   /* @sic R5s170238 change 2.11: single grant sic@ */

    v_MAC_PDU_RX := ?;     // any MAC PDU containing pending data

    v_NB_L2_DATA_IND := f_NBIOT_MAC_CPMode_MacPdu_Receive(v_RLC_CountsInfoListByRef, v_CellId, v_MAC_PDU_RX);

    f_NBIOT_MAC_CPMode_SendRlcStatus(v_RLC_CountsInfoListByRef, v_CellId);

    f_NBIOT_L2_Postamble_State2B_NB(v_CellId, v_RLC_CountsInfoListByRef.AM_VTS, v_RLC_CountsInfoListByRef.AM_VRR);

  }
After change

//New function for 22.3.1.4 step5C
  function fl_TC_22_3_1_4_MacPdu_Receive_step5C(inout NB_RLC_CountsInfoList_Type p_RLC_CountsInfoList,

                                     NBIOT_CellId_Type p_CellId,

                                     template (value) RLC_SDUList_Type p_RlcSduList,

                                     integer p_BSR_Index_Min) runs on NBIOT_PTC

  {

    var template (present) MAC_SDUList_Type v_MAC_SDUList_RX;

    var template (present) MAC_PDU_Type v_MAC_PDU_RX;

    var NB_L2_DATA_IND v_NB_L2_DATA_IND;

    v_MAC_SDUList_RX := { f_NBIOT_MacSdu_RlcAmdPdu_RX(p_RlcSduList) };    /* 2 bytes RLC AM header fixed part, 2 bytes for 1 LI, 7+10 bytes data */

    v_MAC_PDU_RX := (fl_TC_22_3_1_4_MacPdu_CRNTI_BSR_MacSdus(-, -, ?, true, v_MAC_SDUList_RX), 

                     fl_TC_22_3_1_4_MacPdu_CRNTI_BSR_MacSdus(-, -, ?, false, v_MAC_SDUList_RX)); 

    v_NB_L2_DATA_IND := f_NBIOT_MAC_CPMode_MacPdu_Receive(p_RLC_CountsInfoList, p_CellId, v_MAC_PDU_RX);

    fl_MAC_PDU_BSR_CheckGreaterThanOrEqual(v_NB_L2_DATA_IND.L2Data.SubframeData.PduSduList.MacPdu[0], p_BSR_Index_Min);

  }
//New function for 22.3.1.4 step5C
  function fl_TC_22_3_1_4_MacPdu_CRNTI_BSR_MacSdus(B5_Type p_LCID := tsc_LchId_SIB1bis,

                                                   MAC_CTRL_C_RNTI_Type p_C_RNTI := tsc_C_RNTI_Def,

                                                   template (present) B6_Type p_BSR_Value,

                                                   boolean p_C_RNTIFirstFlag := true,

                                                   template (present) MAC_SDUList_Type p_MacSduList := {}) return template (present) MAC_PDU_Type

  {

    var template (present) MAC_Header_Type v_MacPduHeader := {};

    var template MAC_SDUList_Type v_MacSduList := omit;

    var template MAC_CTRL_ElementList_Type v_CtrlElementList := omit;

    var integer v_NoOfSDUs := lengthof(p_MacSduList);

    var B1_Type v_ExtensionBit := tsc_ExtensionMoreHeaders;

    var template (omit) integer v_Length;

    var integer i;

    //C-RNTI,BSR,MAC SDUs

    if(p_C_RNTIFirstFlag)

    {

      v_MacPduHeader := f_NBIOT_MacPduHeader_AddSubheader_RX(v_MacPduHeader, '11011'B, tsc_ExtensionMoreHeaders);

      v_MacPduHeader := f_NBIOT_MacPduHeader_AddSubheader_RX(v_MacPduHeader, tsc_LchId_ShortBSR, tsc_ExtensionMoreHeaders);

    }

    else

    {

      v_MacPduHeader := f_NBIOT_MacPduHeader_AddSubheader_RX(v_MacPduHeader, tsc_LchId_ShortBSR, tsc_ExtensionMoreHeaders);

      v_MacPduHeader := f_NBIOT_MacPduHeader_AddSubheader_RX(v_MacPduHeader, '11011'B, tsc_ExtensionMoreHeaders);

    }

    v_CtrlElementList :={

      ShortBSR               := cr_MAC_BSR_Short(tsc_NBIOT_BSR_LCG, p_BSR_Value),        /* UL only */

      LongBSR                := omit,                                  /* UL only */

      C_RNTI                 := p_C_RNTI,                                  /* UL only */

      ContentionResolutionID := omit,                                  /* DL only */

      TimingAdvance          := omit,

      PowerHeadRoom          := omit,                                   /* UL only */

      ScellActDeact          := omit,        // @sic CA ASP implementation sic@

      ExtPowerHeadRoom       := omit,        // @sic CA ASP implementation sic@

      DC_PowerHeadRoom       := omit

    };

    for (i := 0; i < v_NoOfSDUs; i := i + 1) {

      if (i == v_NoOfSDUs - 1) {

        v_ExtensionBit := tsc_ExtensionNoMoreHeaders;

        v_Length := omit;

      } else {

        v_Length := lengthof(p_MacSduList[i]);

      }

      v_MacPduHeader := f_NBIOT_MacPduHeader_AddSubheader_RX(v_MacPduHeader, p_LCID, v_ExtensionBit, v_Length);

      v_MacSduList[i] := p_MacSduList[i];

    }

    return cr_MAC_PDU(v_MacPduHeader, v_CtrlElementList, v_MacSduList, omit);

  }
  function f_TC_22_3_1_4_NBIOT() runs on NBIOT_PTC

  {

    var NBIOT_CellId_Type v_CellId := nbiot_Cell1;

    var NBIOT_IDLEUPDATED_STATE_Type v_TestLoopMode := STATE2A_NB_TESTLOOP_ModeG;      // !!!! FFS: to be specified in the prose !!!!

    var RetxBSR_Timer_NB_r13 v_RetxBSR_Timer := pp16;

    var PeriodicBSR_Timer_NB_r13 v_PeriodicBSR_Timer := pp64;

    var RetxBSR_Timer_NB_r13 v_RetxBSR_Timer_Step21C := pp64;
    var PeriodicBSR_Timer_NB_r13 v_PeriodicBSR_Timer_Step21C := pp16;
    var template (value) MAC_MainConfig_NB_r13 v_MAC_MainConfig := cs_MAC_MainConfig_NB_SRB;

    var template (value) DL_CCCH_Message_NB v_RrcConnSetup;

    var NB_RLC_CountsInfoList_Type v_RLC_CountsInfoListByRef;                        // RLC status to be maintained by Send/Receive functions

    var SubFrameTiming_Type v_SubFrameTiming;

    var template (value) TimingInfo_Type v_TimingInfo;

    var template (value) RLC_SDUList_Type v_RlcSduList;

    var template (present) MAC_SDUList_Type v_MAC_SDUList_RX;

    var template (present) MAC_PDU_Type v_MAC_PDU_RX; 
    var template (value) NB_RachProcedureList_Type v_RachProcedureList; 
    var NB_SYSTEM_IND v_NB_SYSTEM_IND;
    var NB_L2_DATA_IND v_NB_L2_DATA_IND;

    var integer v_NoOfNpdcchPeriods;

    var integer v_BSR_Index_Min;

    var integer i;

    v_MAC_MainConfig.ul_SCH_Config_r13.periodicBSR_Timer_r13 := v_PeriodicBSR_Timer;

    v_MAC_MainConfig.ul_SCH_Config_r13.retxBSR_Timer_r13 := v_RetxBSR_Timer;

    v_RrcConnSetup := f_NBIOT_508_RRCConnectionSetup(-, -, v_MAC_MainConfig);

    f_NBIOT_L2_Preamble_State2B_NB(v_RLC_CountsInfoListByRef, v_CellId, TransparentMode_RLC_MAC, v_TestLoopMode, v_RrcConnSetup);

    f_NBIOT_SS_ConfigRachPreambleIndMode(v_CellId, enable);
    //------------------------------------

    f_NBIOT_TestBody_Set(true, layer2);

    //@siclog "Step 1 (void)" siclog@

    //@siclog "Step 2" siclog@

    v_SubFrameTiming := f_NBIOT_GetNextSearchSpace(v_CellId, 200);    /* @sic R5s170238 change 2.1: 200ms offset sic@ */

    v_TimingInfo := cs_TimingInfo_SubFrameFull(v_SubFrameTiming);

    v_RlcSduList := { crs_PDCP_SDU_10B, crs_PDCP_SDU_10B };

    f_NBIOT_MAC_CPMode_SendRlcSdus(v_RLC_CountsInfoListByRef, v_CellId, v_TestLoopMode, v_TimingInfo, v_RlcSduList);

    v_BSR_Index_Min := 6;

    //@siclog "Step 3 - 4" siclog@

    v_SubFrameTiming := f_NBIOT_IncrementSearchSpace(v_SubFrameTiming, 3.0);     /* @sic R5-172835/R5s170238 change 2.1: three NPDCCH cycles to give the UE time for processing the data sic@ */

    fl_TC_22_3_1_4_TriggerBSR(v_RLC_CountsInfoListByRef, v_CellId, v_SubFrameTiming, v_BSR_Index_Min);

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 4");


//@siclog "Step 5" siclog@

//@siclog "Step 5B" siclog@


//@siclog "Step 6" siclog@


v_RachProcedureList := { cs_NB_RachProcedure(cs_NB_RandomAccessResponseConfig_Def, cs_NB_ContentionResolutionCtrl_CRNTI(1, 6)) }; //176 bits


f_NBIOT_SS_CommonCellConfig(v_CellId, cads_NB_RachProcedure_Config_REQ(v_CellId, cs_TimingInfo_Now, -, -, v_RachProcedureList));

//@siclog "Step 5A" siclog@


SYSIND.receive(car_NB_RachPreamble_IND(v_CellId)) -> value v_NB_SYSTEM_IND;


f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 5A");

//@siclog "Step 5C" siclog@


v_RlcSduList := { substr(crs_PDCP_SDU_10B, 0, 3)};


fl_TC_22_3_1_4_MacPdu_Receive_step5C(v_RLC_CountsInfoListByRef, v_CellId, v_RlcSduList, 4);//10-3+10=17 betys->Index=4

//@siclog "Step 7" siclog@


v_RlcSduList := { substr(crs_PDCP_SDU_10B, 3, 7), crs_PDCP_SDU_10B};


v_MAC_SDUList_RX := { f_NBIOT_MacSdu_RlcAmdPdu_RX(v_RlcSduList) }; 


v_MAC_PDU_RX := f_NBIOT_MacPdu_BSR_MacSdus_Padding_RX(-, noPadding, noBSR, v_MAC_SDUList_RX); 


f_NBIOT_MAC_CPMode_MacPdu_Receive(v_RLC_CountsInfoListByRef, v_CellId, v_MAC_PDU_RX);


//@siclog "Step 7A" siclog@


f_NBIOT_MAC_CPMode_SendRlcStatus(v_RLC_CountsInfoListByRef, v_CellId);
    //@siclog "Step 8" siclog@

    v_SubFrameTiming := f_NBIOT_GetNextSearchSpace(v_CellId, 200);
    v_TimingInfo := cs_TimingInfo_SubFrameFull(v_SubFrameTiming);

    v_RlcSduList := { crs_PDCP_SDU_10B };

    f_NBIOT_MAC_CPMode_SendRlcSdus(v_RLC_CountsInfoListByRef, v_CellId, v_TestLoopMode, v_TimingInfo, v_RlcSduList);

    //@siclog "Step 9 - 10" siclog@

    v_SubFrameTiming := f_NBIOT_IncrementSearchSpace(v_SubFrameTiming, 3.0);

    v_BSR_Index_Min := 1;

    fl_TC_22_3_1_4_TriggerBSR(v_RLC_CountsInfoListByRef, v_CellId, v_SubFrameTiming, v_BSR_Index_Min);

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 10");

    //@siclog "Step 10A" siclog@

    v_SubFrameTiming := f_NBIOT_IncrementSearchSpace(v_SubFrameTiming, 6.0);
    v_TimingInfo := cs_TimingInfo_SubFrameFull(v_SubFrameTiming);

    f_NBIOT_ULGrantTransmission_ExplicitGrant(v_CellId, 0, 7, v_TimingInfo);  // UL grant of 104 bits (ITBS=7, IRU=0) -> 13 bytes

    //@siclog "Step 10B" siclog@

    v_RlcSduList := { crs_PDCP_SDU_10B };
    v_MAC_SDUList_RX := { f_NBIOT_MacSdu_RlcAmdPdu_RX(v_RlcSduList) };     /* 2 bytes RLC AM header fixed part, 10 bytes data */

    v_MAC_PDU_RX := f_NBIOT_MacPdu_BSR_MacSdus_Padding_RX(-, noPadding, noBSR, v_MAC_SDUList_RX);  /* 1 byte SDU sub-header, 12 bytes MAC SDU */

    f_NBIOT_MAC_CPMode_MacPdu_Receive(v_RLC_CountsInfoListByRef, v_CellId, v_MAC_PDU_RX);


//@siclog "Step 10C" siclog@


f_NBIOT_MAC_CPMode_SendRlcStatus(v_RLC_CountsInfoListByRef, v_CellId);
    //@siclog "Step 11" siclog@

    v_SubFrameTiming := f_NBIOT_GetNextSearchSpace(v_CellId, 200);

    v_TimingInfo := cs_TimingInfo_SubFrameFull(v_SubFrameTiming);

    v_RlcSduList := { crs_PDCP_SDU_10B, crs_PDCP_SDU_10B, crs_PDCP_SDU_10B };

    f_NBIOT_MAC_CPMode_SendRlcSdus(v_RLC_CountsInfoListByRef, v_CellId, v_TestLoopMode, v_TimingInfo, v_RlcSduList);
    //@siclog "Step 12" siclog@

    v_SubFrameTiming := f_NBIOT_IncrementSearchSpace(v_SubFrameTiming, 3.0);
    v_TimingInfo := cs_TimingInfo_SubFrameFull(v_SubFrameTiming);

    f_NBIOT_ULGrantTransmission_ExplicitGrant(v_CellId, 1, 9, v_TimingInfo);  // UL grant of 296 bits (ITBS=9, IRU=1) -> 37 bytes

    //@siclog "Step 13" siclog@

    v_RlcSduList := { crs_PDCP_SDU_10B, crs_PDCP_SDU_10B, crs_PDCP_SDU_10B };

    v_MAC_SDUList_RX := { f_NBIOT_MacSdu_RlcAmdPdu_RX(v_RlcSduList) };     /* 2 bytes RLC AM header fixed part, 3 bytes for 2 LIs, 3 * 10 bytes data */

    v_MAC_PDU_RX := f_NBIOT_MacPdu_BSR_MacSdus_Padding_RX(-, singleBytePadding, noBSR, v_MAC_SDUList_RX);  /* 1 byte padding, 1 byte SDU sub-header, 35 bytes MAC SDU */

    f_NBIOT_MAC_CPMode_MacPdu_Receive(v_RLC_CountsInfoListByRef, v_CellId, v_MAC_PDU_RX);

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 13");

    //@siclog "Step 14" siclog@

    f_NBIOT_MAC_CPMode_SendRlcStatus(v_RLC_CountsInfoListByRef, v_CellId);

    //@siclog "Step 15" siclog@

    v_SubFrameTiming := f_NBIOT_GetNextSearchSpace(v_CellId, 200);    /* @sic R5s170238 change 2.5: 200ms offset sic@ */

    v_TimingInfo := cs_TimingInfo_SubFrameFull(v_SubFrameTiming);

    v_RlcSduList := { crs_PDCP_SDU_8B, crs_PDCP_SDU_7B };

    f_NBIOT_MAC_CPMode_SendRlcSdus(v_RLC_CountsInfoListByRef, v_CellId, v_TestLoopMode, v_TimingInfo, v_RlcSduList);

    //@siclog "Step 16" siclog@

    v_SubFrameTiming := f_NBIOT_IncrementSearchSpace(v_SubFrameTiming, 3.0);     /* @sic R5-172835/R5s170238 change 2.5: three NPDCCH cycles to give the UE time for processing the data sic@ */

    v_TimingInfo := cs_TimingInfo_SubFrameFull(v_SubFrameTiming);

    f_NBIOT_ULGrantTransmission_ExplicitGrant(v_CellId, 1, 8, v_TimingInfo);     /* UL grant of 256 bits (ITBS=8, IRU=1)  @sic R5s170238 change 2.6: ITBS=8 sic@ */

    //@siclog "Step 17" siclog@
    v_MAC_SDUList_RX := { f_NBIOT_MacSdu_RlcAmdPdu_RX(v_RlcSduList) };      /* 2 bytes RLC AM header fixed part, 2 bytes for 1 LI, 8 + 7 bytes data */

    v_MAC_PDU_RX := f_NBIOT_MacPdu_BSR_MacSdus_Padding_RX(-, normalPadding, zeroBSR, v_MAC_SDUList_RX);

    f_NBIOT_MAC_CPMode_MacPdu_Receive(v_RLC_CountsInfoListByRef, v_CellId, v_MAC_PDU_RX);

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 17");
    //@siclog "Step 18" siclog@

    f_NBIOT_MAC_CPMode_SendRlcStatus(v_RLC_CountsInfoListByRef, v_CellId);

    //@siclog "Step 19" siclog@

    v_SubFrameTiming := f_NBIOT_GetNextSearchSpace(v_CellId, 200);
    v_TimingInfo := cs_TimingInfo_SubFrameFull(v_SubFrameTiming);
    v_RlcSduList := { crs_PDCP_SDU_12B };

    f_NBIOT_MAC_CPMode_SendRlcSdus(v_RLC_CountsInfoListByRef, v_CellId, v_TestLoopMode, v_TimingInfo, v_RlcSduList);

    //@siclog "Step 20" siclog@

    v_SubFrameTiming := f_NBIOT_IncrementSearchSpace(v_SubFrameTiming, 3.0);
    v_TimingInfo := cs_TimingInfo_SubFrameFull(v_SubFrameTiming);
    f_NBIOT_ULGrantTransmission_ExplicitGrant(v_CellId, 0, 9, v_TimingInfo);  // UL grant of 136 bits (ITBS=9, IRU=0)

    //@siclog "Step 21" siclog@

    v_MAC_SDUList_RX := { f_NBIOT_MacSdu_RlcAmdPdu_RX(v_RlcSduList) };

    v_MAC_PDU_RX := f_NBIOT_MacPdu_BSR_MacSdus_Padding_RX(-, noPadding, zeroBSR, v_MAC_SDUList_RX);            /* @sic R5s170238 change 2.8 - MMC160 implementation: zeroBSR instead of anyBSR sic@ */

    v_NB_L2_DATA_IND := f_NBIOT_MAC_CPMode_MacPdu_Receive(v_RLC_CountsInfoListByRef, v_CellId, v_MAC_PDU_RX);

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 21");

    //@siclog "Step 21A" siclog@

    f_NBIOT_MAC_CPMode_SendRlcStatus(v_RLC_CountsInfoListByRef, v_CellId);                                     /* @sic R5-172835/R5s170238 change 2.9 sic@ */

    //@siclog "Step 21B" siclog@
    f_NBIOT_SS_ConfigRachPreambleIndMode(v_CellId, disable);
    f_NBIOT_L2_CPMode_RestoreSRBs(v_CellId, v_RLC_CountsInfoListByRef.AM_VTS, v_RLC_CountsInfoListByRef.AM_VRR);

    f_NBIOT_ULGrantTransmission(v_CellId, cs_NB_UL_GrantScheduling_Start);

    f_NBIOT_RRC_ConnectionRelease(v_CellId);
    //@siclog "Step 21C" siclog@
    //'Generic Test Procedure NB-IoT Control Plane CIoT MT user data transfer non-SMS transport' as described in TS 36.508 [18], clause 8.1.5A.2.3 are performed

    v_MAC_MainConfig.ul_SCH_Config_r13.periodicBSR_Timer_r13 := v_PeriodicBSR_Timer_Step21C;

    v_MAC_MainConfig.ul_SCH_Config_r13.retxBSR_Timer_r13 := v_RetxBSR_Timer_Step21C;

    v_RrcConnSetup := f_NBIOT_508_RRCConnectionSetup(-, -, v_MAC_MainConfig);

    f_NBIOT_SS_ConfigRachProcedureMsg4(v_CellId, valueof(v_RrcConnSetup));

    f_NBIOT_508Check_CP_ResponseToPagingForMTAccess(v_CellId);

    f_Delay(0.5); //Wait for SS transmitting DL NAS massage SERVICE ACCEPT
    f_NBIOT_StopULGrantTransmission(v_CellId);                          // stop UL grants

    v_RLC_CountsInfoListByRef := f_NBIOT_SS_RlcCounts_Get(v_CellId);         // get RLC sequence numbers

    f_NBIOT_L2_CPMode_SrbTestMode(v_CellId, TransparentMode_RLC_MAC);
    //@siclog "Step 22" siclog@

    v_RlcSduList := { crs_PDCP_SDU_14B };

    f_NBIOT_MAC_CPMode_SendRlcSdus(v_RLC_CountsInfoListByRef, v_CellId, v_TestLoopMode, -, v_RlcSduList);

    //@siclog "Step 23 - 24" siclog@

    v_SubFrameTiming := f_NBIOT_GetNextSearchSpace(v_CellId);

    v_BSR_Index_Min := 1;

    fl_TC_22_3_1_4_TriggerBSR(v_RLC_CountsInfoListByRef, v_CellId, v_SubFrameTiming, v_BSR_Index_Min);

    for (i := 0; i < 2; i := i + 1) {

      //@siclog "Step 25" siclog@

      v_BSR_Index_Min := 1;

      v_NoOfNpdcchPeriods := fl_TC_22_3_1_4_PeriodicBSR_Timer_NpdcchPeriods(v_PeriodicBSR_Timer_Step21C) + 1;  // "+1" to be sure that PeriodicBSR-Timer has been expired

      v_SubFrameTiming := f_NBIOT_IncrementSearchSpace(v_SubFrameTiming, int2float(v_NoOfNpdcchPeriods));

      //@siclog "Step 26 - 27" siclog@

      fl_TC_22_3_1_4_TriggerBSR(v_RLC_CountsInfoListByRef, v_CellId, v_SubFrameTiming, v_BSR_Index_Min);

      f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 27 - " & int2str(i));

    }

    f_NBIOT_TestBody_Set(false);

    //------------------------------------

    // give default grant to the UE to allow transmission of pending data

    f_NBIOT_ULGrantTransmission(v_CellId, cs_NB_UL_GrantScheduling_Start(cs_NB_SingleGrant));   /* @sic R5s170238 change 2.11: single grant sic@ */

    v_MAC_PDU_RX := ?;     // any MAC PDU containing pending data

    v_NB_L2_DATA_IND := f_NBIOT_MAC_CPMode_MacPdu_Receive(v_RLC_CountsInfoListByRef, v_CellId, v_MAC_PDU_RX);

    f_NBIOT_MAC_CPMode_SendRlcStatus(v_RLC_CountsInfoListByRef, v_CellId);

    f_NBIOT_L2_Postamble_State2B_NB(v_CellId, v_RLC_CountsInfoListByRef.AM_VTS, v_RLC_CountsInfoListByRef.AM_VRR);

  }
